



СЪБИТИЙНО ПРОГРАМИРАНЕ с Visual Basic 6.0

Практическа част


ст. ас. Стефка Анева
Катедра „Компютърни технологии“



Пример 1. Създайте приложение, съдържащо два командни бутона и един етикет. За свойство Caption на етикета задайте стойността “Hello”, а за двата бутона – съответно “On” и “Off”. При позициониране на показалеца на мишката върху един от двата бутона той да се променя съответно във вида 1-Arrow, а при позициониране върху формата – вида на показалеца на мишката да бъде 2- Cross. Съхранете проекта под име project1.vbp, име на формата-form1.frm.




Контрол	Име на обект	Свойства
Label	Label1	Caption="Hello" ; Alignment=2-Centre FontName=Arial ; FontSize=16 BackColor – по избор
Commandbutton	Command1	Caption="On" ; MousePointer=1-Arrow
Commandbutton	Command2	Caption="Off" ; MousePointer=1-Arrow
Form	FrmHello	MousePointer=2-Cross StartPosition=2-CenterScreen



Пример 2. Отворете създадения вече проект project1.vbp и добавете към двата бутона следния програмен код (т.е. активирайте процедурата Click за двата командни бутона):

```
Private Sub Command1_Click ()
    Label1.Visible = True
End Sub
Private Sub Command2_Click ()
    Label1.Visible = False
End Sub
```




Добавете трети бутон към формата със следните свойства:

Контрол	Име на обект	Свойства
Commandbutton	Command3	Caption="Change text"

и следния програмен код:


```
Private Sub Command3_Click ()
    Label1.Caption = InputBox("Enter the new text")
End Sub
```



Пример 3. Създайте приложение за пресмятане на квадратния корен на зададено произволно число. При въвеждане на даденото число в текстово поле, с помощта на команден бутон да се пресмята квадратния корен на числото и получения резултат да се визуализира в етикет. Задайте име на създаденото от вас приложение-Primer3.vbp, име на формата – frmPrimer3.frm.

Контрол	Име на обект	Свойства
Textbox	TxtInput	Text="" Alignment=1-Right Justify
Label	LblResult	Caption="" Alignment=1-Right Justify Border Style=1-Fixed Single BackColor=white
Commandbutton	Cmdres	Caption="Пресмятане на квадратен корен"
Commandbutton	cmdNew	Caption="Нови данни"
Commandbutton	CmdExit	Caption="Изход"
Label	Label1	Caption="Въведи стойност"
Label	Label2	Caption="Резултат"


4/59

Програмен код:


```
Private Sub cmdExit_Click()  
    End  
End Sub  
Private Sub cmdNew_Click()  
    txtInput.Text = ""  
    lblResult.Caption = ""  
    txtInput.SetFocus  
End Sub  
Private Sub cmdRes_Click()  
    lblResult.Caption = Sqr(txtInput.Text)  
End Sub
```

Допълнение: При опит за въвеждане на отрицателно число в текстовото поле да се получава съобщение за грешка.

```
Private Sub cmdRes_Click()  
    If txtInput.Text < 0 Then  
        MsgBox "Въвели сте число, което попада извън дефиниционната област!"  
        Exit Sub  
    Else: lblResult.Caption = Sqr(txtInput.Text)  
    End If  
End Sub
```




Пример 4. Създайте приложение, което с помощта на два командни бутона увеличава и намалява размерите на дадена картинка(действието наподобява пускане и вдигане на щора). Приложението съдържа още два командни бутона, с помощта на които се извършва съответно промяна на картинката и затваряне на формата. Задайте име на проекта – shtora.vbp, име на форма – shtora.frm.



Контрол	Име на обект	Свойства
PictureBox	Picture1	Autosize=True ; Picture – по избор
PictureBox	Picture2	Visible=False ; Picture – по избор
PictureBox	Picture3	Visible=False ; Picture – по избор
PictureBox	Picture4	Visible=False ; Picture – по избор
PictureBox	Picture5	Visible=False ; Picture – по избор
Commandbutton	CmdUp	Caption="Up"
Commandbutton	CmdDown	Caption="Down"
Commandbutton	CmdChange	Caption="Change picture"
Commandbutton	cmdClose	Caption="Close form"

6/59

**Програмен код:**


```
Dim n As Integer
Dim s, p As Integer

Private Sub cmdChange_Click()
    If p = 0 Then
        Picture1.Picture = Picture2.Picture
        p = 1
    ElseIf p = 1 Then
        Picture1.Picture = Picture3.Picture
        p = 2
    ElseIf p = 2 Then
        Picture1.Picture = Picture4.Picture
        p = 3
    Else: Picture1.Picture = Picture5.Picture
        p = 0
    End If
End Sub
```

```
Private Sub cmdClose_Click()
    End
End Sub

Private Sub cmdDown_Click()
    Do While s < n
        Picture1.Height = Picture1.Height + 1
        s = s + 1
    Loop
End Sub


Private Sub cmdUp_Click()
    n = Picture1.Height
    s = n
    Do While s > 50
        Picture1.Height = Picture1.Height - 1
        s = s - 1
    Loop
End Sub
```




Пример 5. Създайте приложение, което съдържа следните елементи (контроли): списъчна кутия, текстово поле, етикет и четири командни бутона. Действието на всеки един от бутоните е следното:

- ◆ При въвеждане на произволна стойност в текстовото поле и натискане на бутон Add да се извършва добавяне на тази стойност в списъчната кутия.
- ◆ Бутон Delete извършва изтриване на маркиран елемент от списъчната кутия.
- ◆ Бутон Clear изтрива всички елементи от списъчната кутия.
- ◆ Бутон Close затваря формата.


Задайте име на проекта – Clientlist.vbp, име на форма – Clientlist.frm.



Контрол	Име на обект	Свойства
Form	frmClientlist	Caption="Client List"
Textbox	txtInput	Text=""
Listbox	lstClientList	
Commandbutton	cmdAdd	Caption="Add"
Commandbutton	cmdDelete	Caption="Delete"
Commandbutton	cmdClear	Caption="Clear"
Commandbutton	cmdClose	Caption="Close"
Label	Labell	Caption="Enter the name to add:"


**Програмен код:**

```
Private Sub cmdAdd_Click()  
    lstClientList.AddItem txtInput.Text  
    txtInput.Text = ""  
    txtInput.SetFocus  
End Sub  
  
Private Sub cmdDelete_Click()  
    If lstClientList.ListIndex >= 0 Then  
        lstClientList.RemoveItem lstClientList.ListIndex  
    Else  
        Beep  
    End If  
End Sub  
  
Private Sub cmdClear_Click()  
    lstClientList.Clear  
End Sub  
  
Private Sub cmdClose_Click()  
    End  
End Sub
```



Пример 6. Създайте приложение, съдържащо следните елементи: списъчна кутия, етикет и текстово поле. При избор на даден елемент от списъчната кутия стойността му да бъде показана и в текстовото поле.

Задайте име на проекта – Fruits.vbp, име на форма – Fruits.frm.




Контрол	Име на обект	Свойства
List box	LstFruits	List=Apples, Pears, Bananas, Lemons
Label	Labell	Caption="Selected Item"
Text box	TxtSelection	Text=""
Form	Form1	Caption="Fruits"

Програмен код:

```
Private Sub lstFruits_Click()
    txtSelection.Text =lstFruits.List(lstFruits.ListIndex)
End Sub
```

10/59




Пример 7. Създайте приложение, което съдържа етикет, текстово поле, три опционални бутона (Option buttons) и две полета за отметка (Checkbox). С помощта на опционалните бутона да се извършва избор на шрифт за текста, въведен в текстовото поле (свойство FontName), както и настройка цвета на фона (свойство BackColor). Посредством полетата за отметка да се избира стойност (True или False) за свойствата FontBold и FontItalic на текстовото поле. Задайте име на проекта - TextFormats.vbp, име на форма - TextFormats.frm.



Контрол	Име на обект	Свойства
Form	Form1	Caption="Text Formats"
Label	Label1	Caption="Enter your text here" Font, FontSize=14
Textbox	txtInput	Text=""
Checkbox	chkBold	Caption="Bold" Value=0-Unchecked
Checkbox	chkItalic	Caption="Italic" Value=0-Unchecked
Optionbutton	optMSSansSerif	Caption="MS Sans Serif" Value=true
Optionbutton	optTimesNewRoman	Caption="Times New Roman"
Optionbutton	optArial	Caption="Arial"

11/59



Програмен код:


```
Private Sub chkBold_Click ()
    If chkBold.Value = 1 Then
        txtInput.FontBold = True
    Else
        txtInput.FontBold = False
    End If
End Sub

Private Sub chkItalic_Click ()
    If chkItalic.Value = 1 Then
        txtInput.FontItalic = True
    Else
        txtInput.FontItalic = False
    End If
End Sub
```

```
Private Sub optMSSansSerif_Click ()
    txtInput.FontName = "MS Sans Serif"
    txtInput.BackColor = vbGreen
End Sub

Private Sub optTimesNewRoman_Click ()
    txtInput.FontName = "Times New Roman"
    txtInput.BackColor = vbRed
End Sub

Private Sub optArial_Click ()
    txtInput.FontName = "Arial"
    txtInput.BackColor = vbBlue
End Sub
```




Пример 8. Създайте приложение, включващо в себе си три форми:

1. Формата List and Combo Demo да бъде стартова форма за проекта (т.е. да се визуализира при стартирането му). Тя съдържа следните елементи: етикет и три командни бутона. Два от тези бутони служат за визуализация на останалите две форми от проекта. Бутон Quit служи за изход от приложението.




2. Формата ClientList.frm (създадена в пример 5) да се визуализира при натискане на бутон List Box от стартовата форма. Тъй като тази форма е вече реализирана, то е необходимо само да бъде добавена към вашия проект.






3. Формата `ComboDemo.frm` да се визуализира при натискане на бутон `Combo Box` от стартовата форма. Тази форма съдържа четири етикета, един бутон (затварящ формата) и три комбинирани кутии, показващи трите различни стила комбинирана кутия (`Combobox`).



Задайте име на проекта – `Demo.vbp`, име на форма – `Demo.frm`.

Елементи на стартовата форма:

Контрол	Име на обект	Свойства
Form	FormDemo	Caption="List and Combo Demo"
Label	Labell	Caption="Select the demonstration of your choice." FontSize=12 Fontname="Arial Black"
Commandbutton	CmdListBox	Caption="List Box"
Commandbutton	CmdCombobox	Caption="Combo Box"
Commandbutton	CmdQuit	Caption="QUIT"

**Програмен код:**


```
Private Sub cmdListBox_Click()
    FrmClientlist.show
End Sub
Private Sub cmdComboBox_Click()
    Combodemo.Show
End Sub
Private Sub cmdQuit_Click()
    end
End Sub
```

Елементи на формата Combodemo.frm:

Контрол	Име на обект	Свойства
Form	ComboDemo	Caption="Combo Demo"
Label	Label1	Caption="Styles of Combo Boxes" FontSize=12 Fontname="Arial Black"
Label	Label2	Caption="Style 0"
Label	Label3	Caption="Style 1"
Label	Label4	Caption="Style 2"
Combobox	Combo1	Style=0 List= eggs, butter, ham
Combobox	Combo2	Style=1 List= eggs, butter, ham
Combobox	Combo3	Style=2 List= eggs, butter, ham
Commandbuton	cmdClose	Caption="CLOSE"

Програмен код:

```
Private Sub cmdClose_Click()
    Unload Me
End Sub
```




Пример 9. Създайте приложение, съдържащо един команден бутон и пет елемента Image. Три от картинките, на които са показани различните състояния на светофара (т.е. свойство Picture – съответно trffc10a.ico; trffc10b.ico; trffc10c.ico) са помощни и при стартирането на проекта не се виждат. За свойство Picture на четвъртия елемент Image е зададен файла car.bmp. Петият елемент Image служи за визуализиране смяната на различните състояния на светофара, която започва след натискането на бутона Change. При състояние на светофара, светещо зелено се стартира движението на колата с цел преминаване през светофара.

Задайте име на проекта – Car.vbp, име на форма – Car.frm.



Контрол	Име на обект	Свойства
Form	frmCar	Caption="The Moving Car"
Command button	cmdChange	Caption="Change"
Image	imgLight	Stretch=True
Image	imgGreen	Stretch=True ; Visible=false Picture=\\...\\trffc10a.ico
Image	imgYellow	Stretch=True ; Visible=false Picture=\\...\\trffc10b.ico
Image	imgRed	Stretch=True ; Visible=false Picture=\\...\\trffc10c.ico
Image	imgCar	Stretch=true ; Picture=cars.bmp




Програмен код:


```
Option Explicit
Dim original As Single

Private Sub Form_Load()
    imglight.Picture = imgYellow.Picture
End Sub

Private Sub cmdChange_Click()
    Select Case imglight.Picture
        Case imgGreen.Picture
            imglight.Picture = imgYellow.Picture
        Case imgYellow.Picture
            imglight.Picture = imgRed.Picture
        Case imgRed.Picture
            imglight.Picture = imgGreen.Picture
    End Select
    original = imgCar.Left
    Do While imgCar.Left > 0
        imgCar.Left = imgCar.Left - 1
    Loop
    imgCar.Left = original
End Sub
```



Пример 10. Създайте приложение, което съдържа два етикета и един Timer control, показващо текущата дата и текущия час, обновявайки го всяка секунда. Задайте име на проекта – Clock.vbp, име на форма – Clock.frm.




Контрол	Име на обект	Свойства
Label	lblTime	FontName=Courier New FontSize=12
Label	lblDate	FontName=Courier New FontSize=10
Timer	Timer1	Enabled=true Interval=1000
Form	Form1	Caption="Clock"

Програмен код:

```
Private Sub Form_Load()
    lblDate.Caption = Date
End Sub
Private Sub Timer1_Timer()
    lblTime.Caption = Time
End Sub
```


18/59




Пример 11. Приложение на Timer control за анимация.

Използвайте вече създаденото приложение в пример 9 и добавете към формата контрол Timer с име `tmrLights`. Преименувайте бутона `cmdChange` на `cmdStartLights` и преместете програмния код от процедурата `cmdChange_Click` в процедурата `tmrLights_Timer`. Двата командни бутона във формата служат за активиране (бутон `StartLights`) и деактивиране на таймера (бутон `StopLights`).

Задайте име на проекта - `CarTimer.vbp`, име на форма - `frmCarTimer.frm`.



Контрол	Име на обект	Свойства
Timer – нов контрол	TmrLights	Enabled=false Interval=1000
Command button – съществуващ контрол	cmdChange (от предишната форма)	Name=cmdStartLights Caption="Start Lights"
Command button – нов контрол	CmdStopLights	Caption="Stop Lights"


**Програмен код:**

```
Option Explicit
Dim original As Single
Dim f As Integer


Private Sub cmdStartLights_Click()
    tmrLights.Enabled = TRUE
End Sub

Private Sub cmdStopLights_Click()
    tmrLights.Enabled = FALSE
End Sub

Private Sub Form_Load()
    imglight.Picture = imgYellow.Picture
    f = 0
End Sub
```



```
Private Sub tmrLights_Timer()  
    Select Case imglight.Picture  
        Case imgGreen.Picture  
            imglight.Picture = imgYellow.Picture  
        Case imgYellow.Picture  
            If f = 0 Then imglight.Picture = imgRed.Picture  
            If f = 1 Then  
                imglight.Picture = imgGreen.Picture  
                original = imgCar.Left  
                Do While imgCar.Left > 0  
                    imgCar.Left = imgCar.Left - 1  
                Loop  
                imgCar.Left = original  
                f = 0  
            End If  
        Case imgRed.Picture  
            imglight.Picture = imgYellow.Picture  
            f = 1  
    End Select  
End Sub
```

**Пример 12.** Анимация, MessageBox, Picture box.

Създайте приложение, съдържащо следните елементи: PictureBox, три командни бутона, две картинки (Image) и един етикет. Елемента PictureBox се явява контейнер за двата командни бутона – Fly и GoBack и за картинката, визуализираща самолет (свойство Picture=plane.ico).


Забележка: Тъй като PictureBox е контейнер, то е необходимо той да бъде нарисован във формата преди другите три елемента(контрола), съдържащи се в него.



При кликване върху картинката, под която има надпис Click Me! се появява следното съобщение:




Непосредствено преди опит за изход от приложението с помощта на бутон Quit да бъде визуализирано следното предупредително съобщение:



Съхранете проекта под име **Flying.vbp**, а формата с име **frmFlying.frm**.

22/59



Контрол	Име на обект	Свойства
Form	Form1	Picture=clouds.bmp
PictureBox	picflying	Picture=clouds.bmp
Commandbutton	cmdfly	Caption="Fly"
Commandbutton	cmdgoback	Caption="Go Back"
Image	imgplane	Picture=plane.ico
Image	imgMe	Picture=...face03.ico
Commandbutton	Quit	Caption="Quit"
Label	lblMe	Caption="Click Me!" FontSize=14


Програмен код:

```


Private Sub cmdFly_Click()
    Do While imgplane.Left < picflying.Width
        imgplane.Left = imgplane.Left + 3
        imgplane.Top = imgplane.Top + 1
    Loop
End Sub
Private Sub cmdGoBack_Click()
    Do While imgplane.Left > 0
        imgplane.Left = imgplane.Left - 3
        imgplane.Top = imgplane.Top - 1
    Loop
End Sub
Private Sub imgme_Click()
    MsgBox "Thanks for clicking on me!", vbInformation, "Info Message"
End Sub
Private Sub Quit_Click()
    If MsgBox("Do you really want to quit?", vbQuestion + vbYesNo + _
        vbDefaultButton2, "Flying") = vbYes Then
        End
    End If
End Sub

```


23/59



Пример 13. Създайте приложение, което съдържа следните елементи: етикет, две рамки (Frames), във всяка от които има група от три опционални бутона (Option buttons). При избор на даден опционален бутон от рамката ForeGround Color се извършва настройка на свойството ForeColor на етикета. При избор на даден опционален бутон от рамката BackGround Color се извършва настройка на свойството BackColor на етикета. Задайте име на проекта – Hello.vbp, име на форма – Hello.frm.




Контрол	Име на обект	Свойства
Frame control	Frame1	Caption="Foreground Color"
Frame control	Frame2	Caption="Background Color"
Form	frmOption	Caption="Group of Option Buttons"
Optionbutton (Control Array)	optForeground	Caption: Red; Yellow; Blue
Label	lblHello	Caption="Hello!" ForeColor=Red; BackColor=Black Alignment=2-Center FontName="Arial"; FontSize=18
Optionbutton (Control Array)	optBackground	Caption: Black; White; Green


**Програмен код:**


```
Private Sub optForeground_Click(Index As Integer)
    Select Case Index
        Case 0: lblHello.ForeColor = vbRed
        Case 1: lblHello.ForeColor = vbYellow
        Case 2: lblHello.ForeColor = vbBlue
    End Select
End Sub

Private Sub optBackground_Click(Index As Integer)
    Select Case Index
        Case 0: lblHello.BackColor = vbBlack
        Case 1: lblHello.BackColor = vbWhite
        Case 2: lblHello.BackColor = vbGreen
    End Select
End Sub
```


**Пример 14. Stopwatch Application**

Създайте приложение, което съдържа три командни бутона и шест етикета. При натискане на бутона StartTiming и EndTiming в етикети се отбелязва съответно времето, в което е бил натиснат всеки един от тези бутона. На базата на стойностите в тези два етикета се получава стойността, която се записва в третия етикет. Тази стойност представлява продължителността във времето(ElapsedTime), започваща с натискането на бутона StartTiming и завършваща с натискането на бутона EndTiming. Бутон Exit служи за изход от приложението. Задайте име на проекта–Stopwatch.vbp, име на формата–Stopwatch.frm.





<u>Контрол</u>	<u>Име на обект</u>	<u>Свойства</u>
Form	FrmStopWatch	Caption="Stop Watch Application" BorderStyle=1-Fixed Single
Commandbutton	cmdStart	Caption=" &Start Timing"
Commandbutton	cmdEnd	Caption=" &End Timing"
Commandbutton	cmdExit	Caption="E&xit"
Label	Label1	Caption="Start Time"
Label	Label2	Caption="End Time"
Label	Label3	Caption="Elapsed Time"
Label	lblStart	Caption="" BorderStyle=1-Fixed Single
Label	lblEnd	Caption="" BorderStyle=1-Fixed Single
Label	lblElapsed	Caption="" BorderStyle=1-Fixed Single

**Програмен код:**


```
Option Explicit
Dim StartTime as Variant
Dim EndTime as Variant
Dim ElapsedTime as Variant

Private Sub cmdStart_click()
    StartTime=Now
    lblStart.Caption=Format(StartTime, "hh:mm:ss")
    lblEnd.Caption=""
    lblElapsed.Caption=""
End sub

Private Sub cmdEnd_click()
    EndTime=Now
    ElapsedTime=EndTime-StartTime
    lblEnd.Caption=Format(EndTime, "hh:mm:ss")
    lblElapsed.Caption=Format(ElapsedTime, "hh:mm:ss")
End sub

Private Sub cmdExit_click()
    end
end sub
```

28/59



Пример 15. Създайте приложение, показващо текущия месец, ден и година. Също така и текущото време, обновявайки го всяка секунда. Прозореца би трябвало да наподобява страница на календар. Задайте име на проекта – Calendar.vbp, име на форма – Calendar.frm.



Контрол	Име на обект	Свойства
Form	frmCalendar	Caption = "My Calendar" BorderStyle = 1 – Fixed Single
Timer	timDisplay	Interval = 1000
Label	lblDay	Caption = "Петък" Font, име на шрифт - MS Sans Serif размер – 24 стил – Bold
Label	lblTime	Caption = "00:00:00 PM" Font, име на шрифт - Times New Roman размер – 24 стил – Bold


29/59



Label	lblYear	Caption = "2003"; Alignment = 2 – Center Font, име на шрифт - Times New Roman размер – 24 стил – Bold
Label	lblNumber	Caption = "11"; Alignment = 2 – Center Font, име на шрифт – Arial размер – 72 стил – Bold
Label	lblMonth	Caption = "Юли"; Alignment = 2 – Center Font, име на шрифт - MS Sans Serif размер – 24 стил – Bold


Програмен код:

```
Private sub timDisplay_Timer()
    Dim Today As Variant
    Today=Now
    lblDay.Caption=Format(Today, "dddd")
    lblMonth.Caption=Format(Today, "mmmm")
    lblYear.Caption=Format(Today, "yyyy")
    lblNumber.Caption=Format(Today, "d")
    lblTime.Caption=Format(Today, "h:mm:ss ampm")
end sub
```




Пример 16. Image Viewer

Създайте приложение, целящо преглеждане на графични файлове (с разширение *.bmp; *.wmf; *.ico). При избор на конкретно устройство, директория и име на файл да се визуализира селектирания файл в елемент Image. Във формата е включен и етикет, който ще показва пълния път за достъп до селектирания файл. Задайте име на проекта–ImgViewer.vbp, име на формата–ImgViewer.frm.




Контрол	Име на обект	Свойства
DriveListbox	Drive1	
DirectoryListbox	Dir1	
FileListbox	File1	Pattern=*.bmp;*.wmf;*.ico
Label	LblFullPath	Caption="Full Path"
Image control	ImgDisplay	Stretch=true
Form	frmImg Viewer	Caption="Image Viewer"


31/59

**Програмен код:**

```
Private Sub Drive1_Change()  
    Dir1.Path = Drive1.Drive  
End Sub  
  
Private Sub Dir1_Change()  
    File1.Path = Dir1.Path  
End Sub  
  
Private Sub File1_Click()  
    Dim Fullpath As String  
    If Right(File1.Path, 1) = "\" Then  
        Fullpath = File1.Path & File1.List(File1.ListIndex)  
    Else  
        Fullpath = File1.Path & "\" & File1.List(File1.ListIndex)  
    End If  
    lblFullpath.Caption = Fullpath  
    imgDisplay.Picture = LoadPicture(Fullpath)  
End Sub
```




Пример 17. Създайте приложение – калкулатор:



Задайте име на проекта–Calculator.vbp, име на формата-Calculator.frm.


Контрол	Име на Обект	Свойства
Label	Label1	Caption="" BorderStyle=1-Fixed Single Alignment=1-Right Justify BackColor=white
Commandbutton	Command1(0), ..., Command1(9) – Control Array	Caption="0".."9"
Commandbutton	Command2	Caption="."
Commandbutton	Command3	Caption="="
Commandbutton	Command4(0..3) – Control Array	Caption="+", "-", "/", "*"
Commandbutton	Command5	Caption="clr"
Form	Калкулатор	Caption="Калкулатор"

33/59

**Програмен код:**


```
Dim result, a(1) As Single
Dim ind As Integer
Dim t As String
Private Sub Command1_Click(Index As Integer)
    Label1.Caption = Label1.Caption +
Command1(Index).Caption
End Sub
Private Sub Command2_Click()
    Label1.Caption = Label1.Caption + Command2.Caption
    Command2.Enabled = False
End Sub
Private Sub Command3_Click()
    a(ind) = Val(Label1.Caption)
    Select Case t
        Case "+"
            result = a(0) + a(1)
        Case "-"
            result = a(0) - a(1)
        Case "*"
            result = a(0) * a(1)
        Case "/"
            If a(1) = 0 Then
                MsgBox "Деление на 0!"
            End If
        Case Else
            result = 0
    End Select
End Sub
```

34/59




```
Else
    result = a(0) / a(1)
End If
End Select
Label1.Caption = result
End Sub
Private Sub Command4_Click(Index As Integer)
    t = Command4(Index).Caption
    Command2.Enabled = True
    a(ind) = Val(Label1.Caption)
    Label1.Caption = ""
    ind = 1
End Sub
Private Sub Command5_Click()
    Label1.Caption = ""
    ind = 0
    Command2.Enabled = True
End Sub
Private Sub Form_Load()
    Label1.Caption = ""
    ind = 0
End Sub
```

35/59



Задача 18. Създайте приложение, което съдържа следните видове елементи на ГПИ: текстово поле (TextBox), етикет (Label), рамки (Frame), командни бутони (CommandButton), опционални бутони (OptionButton), стандартни диалогови прозорци (CommonDialog), изображение (Image) и вертикални ленти за превъртане (VscrollBar). В приложението е необходимо да бъде въведена следната информация: *Име на студента* (Student Name); *Курс* (Grade Level (First, Second, Third, Fourth, Fifth, Sixth)); *Пол* (Student Sex (Male or Female)); *Дата на раждане на студента* (Date of Birth (Month, Day, Year)); *Снимка на студента* (Student Picture). Реализирайте формата на приложението така, че само името на студента да се пише от клавиатурата, а останалата информация да бъде въвеждана посредством изборни бутони (optionbutton), ленти за превъртане (scrollbar) и стандартни диалогови прозорци (common dialog box). Когато формуляра за даден студент бъде попълнен, покажете обобщената информация с помощта на message box. Например:




Задайте име на проекта – studentDB.vbp, име на формата -studentDB.frm.

The diagram illustrates a 'Student Identification' form with the following components and labels:


- txtName**: Text box for the student's name.
- lblMonth**, **lblDay**, **lblYear**: Labels for the birth date fields.
- Frame1**: A frame containing the birth date input fields.
- Frame2**: A frame containing the 'Sex' and 'Grade Level' sections.
- optSex**: Radio buttons for 'Male' and 'Female'.
- optLevel**: Radio buttons for 'First', 'Second', 'Third', 'Fourth', 'Fifth', and 'Sixth' grade levels.
- Frame3**: A frame containing the 'Date of Birth' section.
- vsbMonth**, **vsbDay**, **vsbYear**: Spin buttons for the birth date.
- imgStudent**: A picture box for the student's photo.
- cmdLoad**: A button labeled 'Load Picture'.
- cdlBox**: A common dialog box icon.
- cmdShow**: A button labeled 'Show Student Profile'.
- cmdNew**: A button labeled 'New Student Info'.
- cmdExit**: A button labeled 'Exit'.
- Frame4**: A frame containing the 'Grade Level' section.
- Label1**: A label for the 'Name' field.

The form is titled 'Student Identification' and includes standard window controls (minimize, maximize, close).




Контрол	Име на обект	Свойства
Form	FrmStudent	BorderStyle = 1- Fixed Single Caption = Student Identification
CommandButton	CmdLoad	Caption = &Load Picture
Frame	Frame3	Caption = Picture FontName = MS Sans Serif FontBold = True ; FontSize = 10 FontItalic = True
Image	ImgStudent	BorderStyle = 1 - Fixed Single ; Stretch = True
CommandButton	CmdExit	Caption = E&xit
CommandButton	CmdNew	Caption = &New Student Info
CommandButton	CmdShow	Caption = &Show Student Profile
Frame	Frame4	Caption = Grade Level FontName = MS Sans Serif FontBold = True ; FontSize = 10 FontItalic = True
OptionButton	optLevel	Caption = Sixth ; Index = 5
OptionButton	optLevel	Caption = Fifth ; Index = 4
OptionButton	optLevel	Caption = Fourth ; Index = 3
OptionButton	optLevel	Caption = Third ; Index = 2
OptionButton	optLevel	Caption = Second ; Index = 1
OptionButton	optLevel	Caption = First ; Index = 0
Frame	Frame2	Caption = Sex ; FontName = MS Sans Serif FontBold = True ; FontSize = 10 FontItalic = True

38/59




OptionButton	optSex	Caption = Female ; Index = 1
OptionButton	optSex	Caption = Male ; Index = 0
Frame	Frame1	Caption = Date of Birth FontName = MS Sans Serif ; FontBold = True FontSize = 10 ; FontItalic = True
VScrollBar	vsbYear	Max = 1800 ; Min = 2100 ; Value = 1960
VScrollBar	vsbDay	Max = 1 ; Min = 31 ; Value = 1
VScrollBar	vsbMonth	Max = 1 ; Min = 12 ; Value = 1
Label	lblYear	Alignment = 2 - Center BackColor = &H00FFFFFF & (White) BorderStyle = 1 - Fixed Single FontName = MS Sans Serif FontSize = 10 ; Caption=""
Label	lblDay	Alignment = 2 - Center BackColor = &H00FFFFFF & (White) BorderStyle = 1 - Fixed Single FontName = MS Sans Serif FontSize = 10 ; Caption=""
Label	lblMonth	Alignment = 2 - Center BackColor = &H00FFFFFF & (White) BorderStyle = 1 - Fixed Single FontName = MS Sans Serif FontSize = 10 ; Caption=""
TextBox	txtName	FontName = MS Sans Serif FontSize = 10 ; Text=""
CommonDialog	cdlBox	Filter = *.bmp
Label	Labell	Caption = Name ; FontName = MS Sans Serif FontBold = True FontSize = 10; FontItalic = True


39/59

**Програмен код:**


```
Option Explicit
Dim Months(12) As String
Dim Days(12) As Integer
Dim Grade As String
Private Sub cmdExit_Click()
    End
End Sub
Private Sub cmdLoad_Click()
    cdlbox.ShowOpen
    imgStudent.Picture = LoadPicture(cdlbox.filename)
End Sub
Private Sub cmdNew_Click()
    txtName.Text = ""
    imgStudent.Picture = LoadPicture("")
End Sub
Private Sub cmdShow_Click()
    Dim Is_Leap As Integer
    Dim Msg As String, Age As Integer, Pronoun As String
    Dim M As Integer, D As Integer, Y As Integer
```




```
If vsbMonth.Value = 2 And ((vsbYear.Value Mod 4 = 0 And_
vsbYear.Value Mod 100 <> 0) Or vsbYear.Value Mod 400 = 0) Then
    Is_Leap = 1
Else
    Is_Leap = 0
End If
If vsbDay.Value > Days(vsbMonth.Value) + Is_Leap Then
    MsgBox "Only" + Str(Days(vsbMonth.Value) + Is_Leap) + " days in " +_
Months(vsbMonth.Value), vbOKOnly + vbCritical, "Invalid Birth Date"
    Exit Sub
End If
M = Val(Format(Now, "mm"))
D = Val(Format(Now, "dd"))
Y = Val(Format(Now, "yyyy"))
Age = Y - vsbYear
If vsbMonth.Value > M Or (vsbMonth.Value = M And vsbDay > D) Then Age = Age - 1
```




```
If Age < 0 Then
    MsgBox "Birth date is before current date.", vbOKOnly + _
vbCritical, "Invalid Birth Date"
    Exit Sub
End If
If txtName.Text = "" Then
    MsgBox "The profile requires a name.", vbOKOnly + vbCritical, _
"No Name Entered"
    Exit Sub
End If
Msg = txtName.Text + " is a student in the " + Grade + " grade." + vbCr
If optSex(0).Value = True Then Pronoun = "He " Else Pronoun = "She "
Msg = Msg + Pronoun + " is" + Str(Age) + " years old." + vbCr
MsgBox Msg, vbOKOnly, "Student Profile"
End Sub
```




```
Private Sub Form_Load()  
Months(1) = "January": Days(1) = 31  
Months(2) = "February": Days(2) = 28  
Months(3) = "March": Days(3) = 31  
Months(4) = "April": Days(4) = 30  
Months(5) = "May": Days(5) = 31  
Months(6) = "June": Days(6) = 30  
Months(7) = "July": Days(7) = 31  
Months(8) = "August": Days(8) = 31  
Months(9) = "September": Days(9) = 30  
Months(10) = "October": Days(10) = 31  
Months(11) = "November": Days(11) = 30  
Months(12) = "December": Days(12) = 31  
lblMonth.Caption = Months(vsbMonth.Value)  
lblDay.Caption = Str(vsbDay.Value)  
lblYear.Caption = Str(vsbYear.Value)  
Grade = "first"  
End Sub  
Private Sub optLevel_Click(Index As Integer)  
Select Case Index  
Case 0  
    Grade = "first"
```



```
Case 1
    Grade = "second"
Case 2
    Grade = "third"
Case 3
    Grade = "fourth"
Case 4
    Grade = "fifth"
Case 5
    Grade = "sixth"
End Select
End Sub
Private Sub vsbDay_Change()
    lblDay.Caption = Str(vsbDay.Value)
End Sub
Private Sub vsbMonth_Change()
    lblMonth.Caption = Months(vsbMonth.Value)
End Sub
Private Sub vsbYear_Change()
    lblYear.Caption = Str(vsbYear.Value)
End Sub
```


**Пример 19.** Visual Basic I Access.

Създайте приложение, съдържащо три текстови полета, етикет и един контрол Data. Използваме готова база данни Employees.mdb, съдържаща таблица Employee Details с три полета: FirstName, SurName и Salary. При преглеждане на различните записи в базата данни с помощта на контрола Data в съответните текстови полета се показват данните за всеки запис от таблицата datEmployees. Задайте име на проекта–Datacontrol.vbp, име на формата-frmDatacontrol.frm.




Контрол	Име на обект	Свойства
Data control	DatEmployees	Caption ="Get Next record" DatabaseName=Employees.mdb RecordSource=table"Employee Details"
Label1	LblMessage	
Text1	TxtFirstName	DataSource=datEmployees DataField=FirstName ; Locked=True
Text2	TxtSurname	DataSource=datEmployees DataField=Surname ; Locked=True
Text3	TxtSalary	DataSource=datEmployees DataField=Salary ; Locked=True
Form	FrmDataControl	Caption ="Data Control Demo"

45/59




Програмен код:

```
Private Sub datEmployees_Reposition()  
    Select Case txtSalary.Text  
        Case Is > 40000: lblMessage.Caption = "seriously rich"  
        Case 30000 To 39999: lblMessage.Caption = "almost rich"  
        Case Else: lblMessage.Caption = "has far to go"  
    End Select  
End Sub
```




Пример 20. Работа с готова база данни.

На базата на пример 19 да се добавят към приложението бутони, осъществяващи действията – добавяне, изтриване, търсене и промяна на записи в база данни.




Допълнителни контроли към формата:

Контрол	Име на обект	Свойства
Commandbutton	CmdAdd	Caption="Добавяне"
Commandbutton	CmdDelete	Caption="Изтриване"
Commandbutton	CmdSearch	Caption="Търсене по малко име"
Commandbutton	cmdAmend	Caption="Търсене и промяна"




Програмен код:

```
Private Sub cmdAdd_Click()  
    datEmployees.Recordset.MoveLast  
    datEmployees.Recordset.AddNew  
    datEmployees.Recordset![FirstName] = InputBox("Нови данни: Въведи малко име: ")  
    datEmployees.Recordset![Surname] = InputBox("Нови данни: Въведи фамилия: ")  
    datEmployees.Recordset![Salary] = InputBox("Нови данни: Въведи заплата: ")  
    datEmployees.Recordset.Update  
End Sub  
  
Private Sub cmdAmend_Click()  
    Dim thename, thesearch As String  
    datEmployees.Recordset.MoveFirst  
    thename = InputBox("Who are looking for?")  
    thesearch = "[Firstname]=''" & thename & "'" & ""  
    datEmployees.Recordset.FindFirst thesearch  
    If Not datEmployees.Recordset.NoMatch Then  
        datEmployees.Recordset.Edit  
        datEmployees.Recordset![Salary] = InputBox("Enter new salary")  
        datEmployees.Recordset.Update  
    End If  
End Sub
```



```
Private Sub cmdDelete_Click()  
    datEmployees.Recordset.Delete  
    datEmployees.Recordset.MoveNext  
End Sub  
  
Private Sub cmdSearch_Click()  
    Dim name As String  
    Dim searchName As String  
    name = InputBox("Enter First Name to search")  
    searchName = "[Firstname]= '" & name & "'" & ""  
    datEmployees.Recordset.MoveFirst  
    datEmployees.Recordset.FindFirst searchName  
    If datEmployees.Recordset.NoMatch Then  
        MsgBox "Couldn't find this name"  
    End If  
End Sub  
  
Private Sub datEmployees_Reposition()  
    Select Case txtSalary.Text  
        Case Is > 40000: lblmessage.Caption = "seriously rich"  
        Case 30000 To 39999: lblmessage.Caption = "almost rich"  
        Case Else: lblmessage.Caption = "has far to go"  
    End Select  
End Sub
```

49/59



Пример 21. Създайте приложение, което съдържа следните елементи: един контрол Data, три списъчни кутии(Listbox) и Data bound контролите DBList, DBCombo, DBGrid. Базата данни, която ще използваме в този пример-students.mdb ще бъде създадена в средата на Visual Basic и се състои само от една таблица - stud_info. Контролите DBCombo и DBList са свързани съответно със полетата FirstName и LastName от таблицата stud_info. При селектиране на даден запис в DBCombo и DBList неговата стойност се вписва съответно и в списъчните кутии List1 и List2. При посочване в контрола DBGrid върху определена клетка от даден запис, отразяваща стойността на поле от таблицата, тази селектирана стойност се записва в списъчната кутия List3.

Задайте име на проекта – DataBound.vbp, име на формата – frmDataBound.frm.

Решение:

Първа стъпка:

Най-напред ще създадем нова база данни, като за целта от меню Add-Ins избираме команда Visual Data Manager. От меню File на прозореца VisData избираме команда New, Microsoft Access, Version 7.0 MDB. Задаваме името на новата база данни – students.mdb.



С десен бутон щракаме в прозореца Database Window и от появилото се контекстно меню избираме команда New table. Задаваме име на таблицата – stud_info. Тя се състои от 3 полета (с помощта на бутона AddField се задават всички полета на таблицата):

- ♦ fak_nomer – type Text, Size 7;
- ♦ firstname– type Text, Size 15;
- ♦ lastname– type Text, Size 15;

Най-накрая след задаване на всички полета от таблицата се натиска бутона Build the table.


50/59

Втора стъпка:



Контрол	Име на обект	Свойства
DBCombo	DbCombo1	RowSource=Data1 ListField=FirstName
DBList	DbList1	RowSource=Data1 ListField=LastName DataSource=Data1
Data control	Data1	DatabaseName=student.mdb RecordSource=stud_info
DBGrid	DBGrid1	DataSource=Data1
List box	List1	
List box	List2	
List box	List3	
Form	frmDataBound	Caption="Data Bound Demo"

51/59


**Програмен код:**

```
Private Sub DBCombo1_Click(Area As Integer)
    If Area = dbcAreaList Then
        List1.AddItem DBCombo1.Text
    End If
End Sub


Private Sub DBList1_Click()
    List2.AddItem DBList1.Text
End Sub

Private Sub DBGrid1_MouseUp(Button As Integer, Shift As Integer, _
X As Single, Y As Single)
    Dim theCol As Integer, Bookmark As Variant
    theCol = DBGrid1.ColContaining(X)
    Bookmark=DBGrid1.RowBookmark(DBGrid1.RowContaining(Y))
    List3.AddItem DBGrid1.Columns(theCol).CellValue(Bookmark)
End Sub
```

Допълнение: Аналогично на пример 20 могат да бъдат добавени бутони към формата, реализиращи действията – добавяне, изтриване, търсене и промяна на записи в базата данни.

**Пример 22.** Пример за работа с меню.


Създайте приложение, при което се използва вече създадената форма `frmprimer1.frm` (от пример 3) и към нея се реализира хоризонтално (а след това и вертикално) меню, съдържащо командите – Нови данни, Пресмятане на квадратен корен и Изход.



Указание: Програмният код, съответстващ на процедури `click` за трите командни бутона се записва като програмен код за процедурите `click`, само че отнасящи се за командите от менюто.


Менюто за приложението се състои от три команди:

- Нови данни - *mnuNew*
- Пресмятане на квадратен корен - *mnuRes*
- Изход - *mnuExit*




Програмен код:

```
Private Sub mnuExit_Click()  
    End  
End Sub  
  
Private Sub mnuNew_Click()  
    txtInput.Text = ""  
    lblResult.Caption = ""  
    txtInput.SetFocus  
End Sub  
  
Private Sub mnures_Click()  
    If txtInput.Text < 0 Then  
        MsgBox "Въвели сте число, което попада извън дефиниционната област!"  
        Exit Sub  
    Else: lblResult.Caption = Sqr(txtInput.Text)  
    End If  
End Sub
```




Пример 23. Работа с менюта, Common Dialog control.

Създайте приложение, което съдържа следните елементи на ГПИ: RichTextBox, CommonDialog. Към проекта реализирайте меню, командите за което са описани по-нататък. Приложението има за цел да бъдат реализирани действията, свързани с преглеждане, форматиране и съхраняване на файлове. Задайте име на проекта–File Viewer.vbp, име на формата-File Viewer.frm.




Контрол	Име на обект	Свойства
RichTextbox	rtbDisplay	Scrollbars=3-rtfBOTH
Form	frmFileViewer	Caption="File Viewer"
Common Dialog control	cdlViewer	
Label	Labell	Caption="Full path:" FontSize=10 FontBold=True
Label	lblPath	Caption="" BorderStyle=1-FixedSingle

55/59


**Menu Editor:**

Caption cell	Name cell
&File	mnuFile
&New	mnuNew
-	mnuSep1
&Open File	mnuOpenFile
&Save File	mnuSave
-	mnuSep2
&Exit	mnuExit
Format	mnuFormat
Back Color	mnuBackColor
Change Font	mnuChangeFont
Help	mnuHelp
Rich Text box	mnuRichTextBox
-	mnuSep3
About	mnuAbout

**Програмен код:**


```
Private Sub mnuNew_Click()  
    rtbDisplay.Text = ""  
End Sub  
Private Sub mnuExit_Click()  
    End  
End Sub  
Private Sub mnuAbout_Click()  
    MsgBox "I wrote this little editor program",vbInformation,_  
    "About FileViewer"  
End Sub  
Private Sub mnuOpenFile_Click()  
    Dim strFileName As String  
    cdlViewer.Filter = "RTF|*.rtf|Text|*.txt|All|*.*"  
    cdlViewer.ShowOpen  
    On Error GoTo OpenProblems  
    strFileName = cdlViewer.filename  
    rtbDisplay.LoadFile strFileName  
    lblPath.Caption = strFileName  
    Exit Sub  
OpenProblems:  
    MsgBox "Can't open the file, select another.", vbCritical  
    Exit Sub  
End Sub
```

57/59



```
Private Sub mnuSave_Click()  
    Dim strFileName As String  
    cdlViewer.Filter = "RTF|*.rtf|Text|*.txt"  
    cdlViewer.ShowSave  
    On Error GoTo SaveProblems  
    strFileName = cdlViewer.filename  
    If cdlViewer.FilterIndex = 1 Then  
        cdlViewer.DefaultExt = "RTF"  
        rtbDisplay.SaveFile strFileName, rtfRTF  
    Else  
        cdlViewer.DefaultExt = "txt"  
        rtbDisplay.SaveFile strFileName, rtfText  
    End If  
    Exit Sub  
SaveProblems:  
    MsgBox "Can't save the file, try again.", vbCritical  
    Exit Sub  
End Sub  
  
Private Sub mnuBackColor_Click()  
    cdlViewer.Flags = cdlCCRGBInit  
    cdlViewer.ShowColor  
    rtbDisplay.BackColor = cdlViewer.Color  
End Sub
```

58/59



```
Private Sub mnuChangeFont_Click()  
    cdlViewer.Flags = cdlCFBoth Or cdlCFEffects  
    cdlViewer.ShowFont  
    rtbDisplay.SelFontName = cdlViewer.FontName  
    rtbDisplay.SelFontSize = cdlViewer.FontSize  
    rtbDisplay.SelBold = cdlViewer.FontBold  
    rtbDisplay.SelItalic = cdlViewer.FontItalic  
    rtbDisplay.SelUnderline = cdlViewer.FontUnderline  
    rtbDisplay.SelColor = cdlViewer.Color  
End Sub  
  
Private Sub mnuRichTextBox_Click()  
    cdlViewer.HelpFile = "VB5def.HLP"  
    cdlViewer.HelpCommand = cdlHelpKey  
    cdlViewer.HelpKey = "RichTextbox Control"  
    cdlViewer.ShowHelp  
End Sub
```